Created: 2004-Jul-03

1 Drift operator

1.1 Definition

```
> T:=proc(Index,power,operand)
> local y,t,x,i,Y;
> x:= [dsolve({diff(y(t),t) = subs(Index=y(t),power), y(0)=Index},
> y(t))];
> x:=allvalues(x);
> Y:=rhs(x[1]);
> for i from 2 to nops(x) do
> Y:= piecewise(simplify(subs({t=0},rhs(x[i])))=Index,rhs(x[i]),Y);
> end do;
> subs(Index=Y, operand):
> subs(t=1,%);
> simplify(%);
> end:
```

1.2 Examples

```
> T(x,a,V(x));
```

$$\mathrm{V}(a + x)$$

```
> T(y,a,V(x*y));
```

$$\mathrm{V}(x\,(a + y))$$

```
> T(x,a+b*x,V(x));
```

$$\mathrm{V}(-\frac{a}{b} + \frac{e^b\,(a + b\,x)}{b})$$

```
> T(x,a/x,x);
```

$$\begin{cases} -\sqrt{2\,a + x^2} & -\mathrm{csgn}(x)\,x = x \\ \sqrt{2\,a + x^2} & \mathit{otherwise} \end{cases}$$

```
> T(x, sqrt(x),V(x));
```

$$\mathrm{V}(\frac{1}{4} + \sqrt{x} + x)$$

```
> T(x, a*exp(b*x), V(x));
```

$$\mathrm{V}\left(\frac{\ln\left(-\frac{1}{a\,b\,(1 - \frac{1}{e^{(b\,x)}\,a\,b})}\right)}{b}\right)$$

2 Blur operator

2.1 Definition

```
> stdblur:= proc(Index, operand)
> local u;
> if type(operand,list) then
> map(V->1/sqrt(2*Pi)*int(exp(-u^2/2)*subs(Index=Index-u, V),
> u=-infinity..infinity),
> operand);
> else
> 1/sqrt(2*Pi)*int(exp(-u^2/2)*subs(Index=Index-u, operand),
> u=-infinity..infinity);
> end if;
> simplify(%);
> end:
```

```
> B:=proc(Index, power, operand, check)
> local Op1, Op2, y;
> Op1:= T(Index, power*y, Index);
> Op2:= T(Index, power*y, operand);
> Op1:= stdblur(y, Op1);
> Op2:= stdblur(y, Op2);
> T(y, subs(y=0,Index-Op1)/diff(Op1,y),Op2);
> subs(y=0,%);
> simplify(%);
> end:
```

2.2 Examples

```
> B(x,sigma,a*x^2+b*x+c);
```

$$a\,x^2 + b\,x + c + a\,\sigma^2$$

```
> B(x,sigma,x^5);
```

$$x\,(x^4 + 10\,\sigma^2\,x^2 + 15\,\sigma^4)$$

```
> B(x,x*sigma, [x^(-2),x^(-1),x,x^2,x^3,x^4,x^5]):
> simplify(%);
```

$$[\frac{e^{(3\,\sigma^2)}}{x^2},\, \frac{e^{(\sigma^2)}}{x},\, x,\, e^{(\sigma^2)}\,x^2,\, e^{(3\,\sigma^2)}\,x^3,\, e^{(6\,\sigma^2)}\,x^4,\, e^{(10\,\sigma^2)}\,x^5]$$

```
> B(x,sigma, exp(a*x));
```

$$e^{(\frac{a\,(2\,x+a\,\sigma^2)}{2})}$$

```
> B(x, sigma, V(x));
```

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,\mathrm{V}(-\sigma\,u + x)\,du\right)$$

```
> B(x, x*sigma, V(x));
```

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,\mathrm{V}(x\,e^{(-\frac{\sigma\,(2\,u+\sigma)}{2})})\,du\right)$$

```
> assume(z>0);assume(z2>0);
> B(z, sqrt(z), V(z));
```

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,\mathrm{V}(z\tilde{} - \frac{1}{4} - \frac{u\,\sqrt{4\,z\tilde{} - 1}}{2} + \frac{u^2}{4})\,du\right)$$

```
> B(z,sqrt(z)*sqrt(z2), z^2);
> diff(%,z2);
> diff(%%,z,z);
```

$$z\tilde{}^2 + z2\tilde{}\, z\tilde{} - \frac{1}{8}\, z2\tilde{}^2$$

$$z\tilde{} - \frac{z2\tilde{}}{4}$$

$$2$$

3 Stock price models

3.1 Levy process

```
> L:=proc(Index, power, operand)
> local kernel,u;
> int(exp(I*u*Index + power), Index=-infinity..infinity);
> kernel:= simplify(%);
> int(kernel*subs(Index=Index-u,operand),u=-infinity..infinity);
> simplify(%/(2*Pi));
> end proc:
```

```
> L(x,-2*x*x/2,x^2);
```

$$x^2 + 2$$

```
> L(x,-x*x/2, L(x, -x*x/2, x^2));
```

$$x^2 + 2$$

3.2 Black&Scholes

```
> Vcall:=charfcn[0..infinity](S-K)*(S-K);
```

$$Vcall := charfcn_{0..\infty}(S - K)\,(S - K)$$

```
> Theta:=(V,M)->exp(-r*M)*T(S,r*S*M, B(S,sigma*S*sqrt(M),V));
```

$$\Theta := (V,\, M) \to e^{(-r\,M)}\, \mathrm{T}(S,\, r\,S\,M,\, \mathrm{B}(S,\, \sigma\,S\,\sqrt{M},\, V))$$

```
> F:=Theta(Vcall,M);
```

$$F := \frac{1}{2}\left(\frac{e^{(-r\,M)}\,\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} (S\,e^{(r\,M)}\,e^{(-\frac{\sigma\,(\sigma\,M+2\,u\,\sqrt{M})}{2})} - K)\,e^{(-\frac{u^2}{2})}\,charfcn_{0..\infty}(S\,e^{(r\,M)}\,e^{(-\frac{\sigma\,(\sigma\,M+2\,u\,\sqrt{M})}{2})} - K)\,du\right)$$

```
> subs({M=5, r=5/100, S=100, sigma=4/10, K=100}, F);
```

$$\frac{1}{2}\left(\frac{e^{(-1/4)}\,\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} (100\,e^{(1/4)}\,e^{(-2/5-\frac{2\,u\,\sqrt{5}}{5})} - 100)\,e^{(-\frac{u^2}{2})}\,charfcn_{0..\infty}(100\,e^{(1/4)}\,e^{(-2/5-\frac{2\,u\,\sqrt{5}}{5})} - 100)\,du\right)$$

```
> evalf(%);
```

$$42.87636389$$

3.3 Asian Option

```
> C:=charfcn[0..infinity](A/n-K);
```

$$C := charfcn_{0..\infty}(\frac{A}{n} - K)$$

```
> T(A,S,C);
```

$$charfcn_{0..\infty}(-\frac{-S - A + K\,n}{n})$$

```
> B(S, sigma*S, %);
```

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,charfcn_{0..\infty}\left(-\frac{-S\,e^{(-\frac{u}{4}-1/16)} + S\,e^{(-\frac{u}{4}-1/32)} - A\,e^{(\frac{-1}{32})} + A + K\,n\,e^{(\frac{-1}{32})} - K\,n}{(e^{(\frac{-1}{32})} - 1)\,n}\right) du\right)$$

```
> T(A,S,%);
```

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,charfcn_{0..\infty}\left(-\frac{-S\,e^{(-\frac{u}{4}-1/16)} + S\,e^{(-\frac{u}{4}-1/32)} - S\,e^{(\frac{-1}{32})} - A\,e^{(\frac{-1}{32})} + S + A + K\,n\,e^{(\frac{-1}{32})} - K\,n}{(e^{(\frac{-1}{32})} - 1)\,n}\right) du\right)$$

```
> #B(S,sigma*S, %);
```

3.4 Geometric Brown

3.4.1 Desired result

```
> B(x, sqrt(t)*sigma*x, f(x));
```

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,\mathrm{f}(x\,e^{(-\frac{\sigma\,(2\,u\,\sqrt{t}+\sigma\,t)}{2})})\,du\right)$$

3.4.2 Equivalent operator sequence

```
> T(x,x*x2,f(x));
> B(x2,sigma*sqrt(t),%);
> simplify(T(x,-x*(x2+sigma^2*t/2),%));
```

$$\mathrm{f}(x\,e^{x2})$$

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,\mathrm{f}(x\,e^{(-\sigma\,\sqrt{t}\,u+x2)})\,du\right)$$

$$\frac{1}{2}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\int_{-\infty}^{\infty} e^{(-\frac{u^2}{2})}\,\mathrm{f}(x\,e^{(-\frac{\sigma\,(\sigma\,t+2\,u\,\sqrt{t})}{2})})\,du\right)$$

3.4.3 Proof

```
> diff(T(x,-x*(x2+t*sigma^2/2),f(x,t)),t);
```

$$-\frac{1}{2}\,\mathrm{D}_1(f)(x\,e^{(-x2-\frac{\sigma^2\,t}{2})},\,t)\,x\,\sigma^2\,e^{(-x2-\frac{\sigma^2\,t}{2})}+\mathrm{D}_2(f)(x\,e^{(-x2-\frac{\sigma^2\,t}{2})},\,t)$$

```
> diff(T(x,x*x2,f(x)),x);
```

$$\mathrm{D}(f)(x\,e^{x2})\,e^{x2}$$

```
> B(x2,sigma,exp(x2)*f(x));
```

$$e^{(x2+\frac{\sigma^2}{2})}\,\mathrm{f}(x)$$

```
> diff(T(x,x*x2,f(x)),x2,x2);
```

$$(\mathrm{D}^{(2)})(f)(x\,e^{x2})\,x^2\,(e^{x2})^2+\mathrm{D}(f)(x\,e^{x2})\,x\,e^{x2}$$

4 Intertemporal optimization

4.1 Model selection

Execute only one of the following models

```
> Theta:=V->B(x,1/20,T(x,1/10,V));
> U:=-(x-1)^2;
```

$$\Theta := V \to \mathrm{B}(x,\,\frac{1}{20},\,\mathrm{T}(x,\,\frac{1}{10},\,V))$$

$$U := -(x-1)^2$$

```
> Theta:=V->B(x,1/5,T(x,1/10,V));
> U:=-exp(-x);
```

$$\Theta := V \to \mathrm{B}(x,\,\frac{1}{5},\,\mathrm{T}(x,\,\frac{1}{10},\,V))$$

$$U := -e^{(-x)}$$

```
> Theta:=V->B(x,1/5,T(x,1/10,T(c,1/10*c,V)));
> U:=-exp(-x);
```

$$\Theta := V \to \mathrm{B}(x,\,\frac{1}{5},\,\mathrm{T}(x,\,\frac{1}{10},\,\mathrm{T}(c,\,\frac{c}{10},\,V)))$$

4.2 Optimization

```
> plot({U,Theta(U)},x=-0..2);
```




```
> A:=V->T(h,n,T(c,-n*x,V));
```

$$A := V \to \mathrm{T}(h,\, n,\, \mathrm{T}(c,\, -n\,x,\, V))$$

```
> Da:=V->subs({c=0,h=0},V);
```

$$Da := V \to \mathrm{subs}(\{c = 0,\, h = 0\},\, V)$$

```
> V0:=[subs(x=c+h*x,U),h*x+c];
```

$$V0 := [-e^{(-c-h\,x)},\, c + h\,x]$$

```
> V0:=collect(Theta(V0),h);
> V0:=collect(A(V0),h);
> diff(V0[1],n);
> star:=solve(diff(V0[1],n)=0,n);
```

$$V0 := [-e^{(-c-1/10\,h-h\,x+1/50\,h^2)},\, c + h\,(\frac{1}{10} + x)]$$

$$V0 := [-e^{(-c-\frac{n}{10}-\frac{h}{10}-h\,x+\frac{n^2}{50}+\frac{n\,h}{25}+\frac{h^2}{50})},\, h\,(\frac{1}{10} + x) + c + \frac{n}{10}]$$

$$-(-\frac{1}{10} + \frac{n}{25} + \frac{h}{25})\, e^{(-c-\frac{n}{10}-\frac{h}{10}-h\,x+\frac{n^2}{50}+\frac{n\,h}{25}+\frac{h^2}{50})}$$

$$star := \frac{5}{2} - h$$

```
> plot(subs(x=1,Da(V0)),n);
```




```
> V0:=subs(n=star,V0):
> collect(expand(V0),x);
```

$$[-\frac{h^2\,x^2}{125} + (\frac{2}{125}\,h - \frac{2}{125}\,c\,h)\,x - \frac{c^2}{125} + \frac{2\,c}{125} - \frac{1}{125},\ \frac{c}{125} + \frac{h\,x}{125} + \frac{124}{125}]$$

```
> Da(V0);
> evalf(subs(x=1,%));
```

$$[\frac{-1}{125},\ \frac{124}{125}]$$

$$[-0.008000000000,\ 0.9920000000]$$

5 Statistical measures

5.1 Model selection

```
> Theta:=V->T(x,1,B(x,sigma,V));
```

$$\Theta := V \to \mathrm{T}(x,\ 1,\ \mathrm{B}(x,\ \sigma,\ V))$$

```
> Theta:=V->B(x,sigma,B(y,sigma,T(x,y,V)));
```

$$\Theta := V \to \mathrm{B}(x,\ \sigma,\ \mathrm{B}(y,\ \sigma,\ \mathrm{T}(x,\ y,\ V)))$$

```
> Theta:=V->T(x,y,B(x,sigma,B(y,sigma,T(x,-y,V))));
```

$$\Theta := V \to \mathrm{T}(x,\ y,\ \mathrm{B}(x,\ \sigma,\ \mathrm{B}(y,\ \sigma,\ \mathrm{T}(x,\ -y,\ V))))$$

```
> Theta:=V->T(x,y+1,B(x,sigma,B(y,sigma,T(x,-y,V))));
```

$$\Theta := V \to \mathrm{T}(x,\ y+1,\ \mathrm{B}(x,\ \sigma,\ \mathrm{B}(y,\ \sigma,\ \mathrm{T}(x,\ -y,\ V))))$$

5.2 Expectation value

```
> ev:=x->Theta(x);
```

$$ev := \Theta$$

```
> ev(x);
> ev(y);
```

$$1 + x$$

$$y$$

5.3 Variance

```
> var:=x->simplify(Theta(x^2)-Theta(x)^2);
```

$$var := x \to \mathrm{simplify}(\Theta(x^2) - \Theta(x)^2)$$

```
> var(x);
> var(y);
```

$$2\,\sigma^2$$

$$\sigma^2$$

5.4 Covariance

```
> cov:=(x,y)->simplify(Theta(x*y)-Theta(x)*Theta(y));
```

$$cov := (x,\, y) \to \mathrm{simplify}(\Theta(x\,y) - \Theta(x)\,\Theta(y))$$

```
> cov(x,y);
```

$$-\sigma^2$$

6 Term structure models

6.1 Hoo and Lee

6.1.1 Initial parameter values

```
> unassign('sigma');
> myvalues:={a1=0, a2=0, c=100, t=0, r=1/10,sigma=1/10};
> da := V->evalf(subs(myvalues,V));
```

$$myvalues := \{a1 = 0,\, a2 = 0,\, c = 100,\, t = 0,\, r = \frac{1}{10},\, \sigma = \frac{1}{10}\}$$

$$da := V \to \mathrm{evalf}(\mathrm{subs}(myvalues,\, V))$$

6.1.2 Forward rate curve

```
> f:= M-> 1/10 + 0*M + a1 + a2*M;
```

$$f := M \to \frac{1}{10} + a1 + a2\,M$$

6.1.3 Theta operator

```
> Theta := proc(V)
> B(a1, sigma,
> T(a2, mu2,
> T(a1, mu1,
> T(eur, -f(t)*eur,
> T(t,1,
> V
> )))));
> simplify(%);
> end proc:
```

6.1.4 Fitting the drift condition

```
> unassign('mu1');
> unassign('mu2');
> unassign('c');
```

```
> Theta(eur):
> simplify(%):
> L1:=da(%);
```

$$L1 := e^{(-0.09500000000-1.\,\mu 1)}\,eur$$

```
> Theta(Theta(eur)):
> simplify(%):
> L2:=da(%);
```

$$L2 := e^{(-0.1750000000-3.\,\mu 1-2.\,\mu 2)}\,eur$$

```
> L:=da(solve(
> {L1=eur*exp(-f(0)),
> L2=eur*exp(-f(0)-f(1))}
> , {mu2,mu1}));
```

$$L := \{\mu 2 = 0.005000000000,\ \mu 1 = 0.005000000000\}$$

```
> assign(L);
```

6.1.5 Examples

```
> da(Theta(Theta(Theta(eur))));
```

$$0.7408182207\,eur$$

```
> 100*exp(-f(0)-f(1)-f(2));
> da(%);
```

$$100\,e^{(-3/10-3\,a1-3\,a2)}$$

$$74.08182207$$

```
> f(M):
> da(%);
```

$$0.1000000000$$

```
> Theta(%%):
> da(%);
```

$$0.1200000000 + 0.02000000000\,M$$

6.2 Hull and White

6.2.1 Volatility term structure

```
> assume(kappa>0);
> sigma:=alpha*exp(-kappa*(M-t));
```

$$\sigma := \alpha\,e^{(-\kappa\tilde{}\,(M-t))}$$

6.2.2 Initial values

```
> X1:={h1=0,h2=0,t=0,alpha=.2,kappa=.05,c=0};
```

$$X1 := \{h1 = 0,\ h2 = 0,\ \kappa\tilde{} = 0.05,\ c = 0,\ \alpha = 0.2,\ t = 0\}$$

```
> da:=V->
> simplify(subs(X1,V));
```

$$da := V \rightarrow \mathrm{simplify}(\mathrm{subs}(X1,\ V))$$

```
> daf:=V->evalf(da(V));
```

$$daf := V \rightarrow \mathrm{evalf}(\mathrm{da}(V))$$

6.2.3 Basis functions

```
> iprod:=(f,g)->int(f*g,M=0..infinity);
```

$$iprod := (f,\, g) \to \int_0^\infty f\, g\, dM$$

```
> unassign('k');
> sigma0:=subs(t=0,sigma):
> b1:=k[0]*sigma0;
> b2:=k[1]*sigma0 + k[2]*sigma0*int(sigma0,M);
```

$$b1 := k_0\, \alpha\, e^{(-\kappa\tilde{}\; M)}$$

$$b2 := k_1\, \alpha\, e^{(-\kappa\tilde{}\; M)} - \frac{k_2\, \alpha^2\, (e^{(-\kappa\tilde{}\; M)})^2}{\kappa\tilde{}}$$

```
> S:=solve({
> iprod(b1,b1)=1,
> iprod(b1,b2)=0,
> iprod(b2,b2)=1}, {k[0],k[1],k[2]});
```

$$S := \{k_1 = \frac{4\, \mathrm{RootOf}(-\kappa\tilde{} + \_Z^2,\, label = \_L4)}{\alpha},\, k_2 = \frac{6\, \mathrm{RootOf}(-\kappa\tilde{} + \_Z^2,\, label = \_L4)\, \kappa\tilde{}}{\alpha^2},\, k_0 = \frac{\mathrm{RootOf}(-2\, \kappa\tilde{} + \_Z^2,\, label = \_L5)}{\alpha}\}$$

```
> allvalues(S);
```

$$\left\{k_1 = \frac{4\sqrt{\kappa\tilde{}}}{\alpha},\, k_2 = \frac{6\, \kappa\tilde{}^{(3/2)}}{\alpha^2},\, k_0 = \frac{\sqrt{2}\sqrt{\kappa\tilde{}}}{\alpha}\right\},\, \left\{k_1 = -\frac{4\sqrt{\kappa\tilde{}}}{\alpha},\, k_2 = -\frac{6\, \kappa\tilde{}^{(3/2)}}{\alpha^2},\, k_0 = \frac{\sqrt{2}\sqrt{\kappa\tilde{}}}{\alpha}\right\},$$
$$\left\{k_1 = \frac{4\sqrt{\kappa\tilde{}}}{\alpha},\, k_2 = \frac{6\, \kappa\tilde{}^{(3/2)}}{\alpha^2},\, k_0 = -\frac{\sqrt{2}\sqrt{\kappa\tilde{}}}{\alpha}\right\},$$
$$\left\{k_1 = -\frac{4\sqrt{\kappa\tilde{}}}{\alpha},\, k_2 = -\frac{6\, \kappa\tilde{}^{(3/2)}}{\alpha^2},\, k_0 = -\frac{\sqrt{2}\sqrt{\kappa\tilde{}}}{\alpha}\right\}$$

```
> assign(%[1]);
```

```
> b1;
> b2;
```

$$\sqrt{2}\sqrt{\kappa\tilde{}}\, e^{(-\kappa\tilde{}\; M)}$$

$$4\sqrt{\kappa\tilde{}}\, e^{(-\kappa\tilde{}\; M)} - 6\sqrt{\kappa\tilde{}}\, (e^{(-\kappa\tilde{}\; M)})^2$$

```
> iprod(b1,b1);
> iprod(b1,b2);
> iprod(b2,b2);
```

$$1$$
$$0$$
$$1$$

```
> s1:=iprod(sigma, b1);
> s2:=iprod(sigma, b2);
```

$$s1 := \frac{1}{2}\, \frac{\alpha\sqrt{2}\, e^{(\kappa\tilde{}\; t)}}{\sqrt{\kappa\tilde{}}}$$

$$s2 := 0$$

6.2.4 Yield curve

```
> f00:=1/10+1/100*M;
```

$$f00 := \frac{1}{10} + \frac{M}{100}$$

```
> f0:=f00+h1*b1+h2*b2;
```

$$f0 := \frac{1}{10} + \frac{M}{100} + h1\,\sqrt{2}\,\sqrt{\kappa\tilde{}}\,e^{(-\kappa\tilde{}\,M)} + h2\,(4\,\sqrt{\kappa\tilde{}}\,e^{(-\kappa\tilde{}\,M)} - 6\,\sqrt{\kappa\tilde{}}\,(e^{(-\kappa\tilde{}\,M)})^2)$$

```
> int(subs(M=M2,f0),M2=t..M);
```

$$\frac{1}{200}(20\,M\,\sqrt{\kappa\tilde{}} + M^2\,\sqrt{\kappa\tilde{}} - 200\,h1\,\sqrt{2}\,e^{(-\kappa\tilde{}\,M)} - 800\,h2\,e^{(-\kappa\tilde{}\,M)} + 600\,h2\,(e^{(-\kappa\tilde{}\,M)})^2$$
$$- 20\,t\,\sqrt{\kappa\tilde{}} - t^2\,\sqrt{\kappa\tilde{}} + 200\,h1\,\sqrt{2}\,e^{(-\kappa\tilde{}\,t)} + 800\,h2\,e^{(-\kappa\tilde{}\,t)} - 600\,h2\,(e^{(-\kappa\tilde{}\,t)})^2)\Big/$$
$$\sqrt{\kappa\tilde{}}$$

```
> Z:=M->-1/200*(-20*M*kappa^(1/2)-M^2*kappa^(1/2)+200*h1*2^(1/2)*exp(-k
> appa*M)+800*h2*exp(-kappa*M)-600*h2*exp(-kappa*M)^2+20*t*kappa^(1/2)+t
> ^2*kappa^(1/2)-200*h1*2^(1/2)*exp(-kappa*t)-800*h2*exp(-kappa*t)+600*h
> 2*exp(-kappa*t)^2)/kappa^(1/2);
```

$$Z := M \to -\frac{1}{200}(-20\,M\,\sqrt{\kappa} - M^2\,\sqrt{\kappa} + 200\,h1\,\sqrt{2}\,e^{(-\kappa\,M)} + 800\,h2\,e^{(-\kappa\,M)}$$
$$- 600\,h2\,(e^{(-\kappa\,M)})^2 + 20\,t\,\sqrt{\kappa} + t^2\,\sqrt{\kappa} - 200\,h1\,\sqrt{2}\,e^{(-\kappa\,t)} - 800\,h2\,e^{(-\kappa\,t)}$$
$$+ 600\,h2\,(e^{(-\kappa\,t)})^2)\Big/\sqrt{\kappa}$$

6.2.5 Risk neutral drift

```
> mu:=sigma*int(subs(M=M2,sigma),M2=t..M):
> simplify(%);
> m1:=iprod(mu,b1);
> m2:=iprod(mu,b2);
```

$$-\frac{\alpha^2\,e^{(\kappa\tilde{}\,(-M+t))}\,(e^{(\kappa\tilde{}\,(-M+t))} - 1)}{\kappa\tilde{}}$$

$$m1 := -\frac{1}{6}\,\frac{\alpha^2\,e^{(\kappa\tilde{}\,t)}\,\sqrt{2}\,(2\,e^{(\kappa\tilde{}\,t)} - 3)}{\kappa\tilde{}^{(3/2)}}$$

$$m2 := \frac{1}{6}\,\frac{(e^{(\kappa\tilde{}\,t)})^2\,\alpha^2}{\kappa\tilde{}^{(3/2)}}$$

```
> simplify(m1*b1+m2*b2-mu);
```

$$0$$

6.2.6 Theta Operator

```
> Theta:=V->
> B(h1, s1,
> T(h1, m1,
> T(h2, m2,
> T(eur,eur*(-Z(t+1)+Z(t)),
> T(t,1,
> V
> )))));
```

$$\Theta := V \to \mathrm{B}(h1,\,s1,\,\mathrm{T}(h1,\,m1,\,\mathrm{T}(h2,\,m2,\,\mathrm{T}(eur,\,eur\,(-\mathrm{Z}(t+1) + \mathrm{Z}(t)),\,\mathrm{T}(t,\,1,\,V)))))$$

```
> Theta(eur):
> da(%);
```

$$0.9003245226\, eur$$

```
> Theta(%%):
> daf(%);
```

$$0.8025187980\, eur$$

6.2.7 Expected interest rates

```
> F:=diff(exp(Z(M)),M)/exp(Z(M)):
> da(%);
```

$$0.1000000000 + 0.01000000000\, M$$

```
> Theta(F):
> seq(da(subs(M=i,%)),i=0..10);
```

0.1000000000, 0.07288639480, 0.05111466810, 0.03408819550, 0.02127143430, 0.01218390130, 0.006394732300, 0.003517771200, 0.003207134500, 0.005153206500, 0.009079025200

6.2.8 Cash account

```
> F:=100*eur;
> da(F);
```

$$F := 100\, eur$$

$$100.\, eur$$

```
> F:=Theta(F):
> daf(F);
```

$$74.08182207\, eur$$

```
> seq(100*daf(exp(-Z(i)+Z(0))),i=0..10);
```

100., 90.03245226, 80.25187980, 70.82203535, 61.87833918, 53.52614285, 45.84060113, 38.86795709, 32.62797946, 27.11725350, 22.31301601

6.2.9 Strategy

```
> F:=T(c,eur,Theta(c+eur)):
> da(%);
```

$$1.900324523\, eur$$

```
> P:= V->
> T(c,-a*0.8*eur,
> Theta(
> T(c,a*eur,
> V
> ))):
> da(%);
```

$$V \to \mathrm{T}(0,\, -0.8\, a\, eur,\, \Theta(\mathrm{T}(0,\, a\, eur,\, V)))$$

```
> M_1:=da(P(c));
```

$$M\_1 := 0.1003245226\, a\, eur$$

```
> M_2:=da(P(c^2));
```

$$M\_2 := 0.04150802339\, a^2\, eur^2$$

```
> sqrt(M_2-M_1^2);
```

$$0.1773217797\, \sqrt{a^2\, eur^2}$$

```
> #VR:=P(charfcn[-infinity..u](c)):
```